Rules for integrands of the form $(c + d x)^m (a + b \operatorname{Sec}[e + f x])^n$

1. $\int (c + d x)^m (b \operatorname{Sec}[e + f x])^n \, dx$

   1. $\int (c + d x)^m \operatorname{Sec}[e + f x]^n \, dx$ when $n > 0$

      1: $\int (c + d x)^m \operatorname{Sec}[e + f x] \, dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\operatorname{Csc}[e + f x] == -\partial_x \frac{2 \operatorname{ArcTanh}\left[e^{i (e+f x)}\right]}{f} == \partial_x \frac{\operatorname{Log}\left[1-e^{i (e+f x)}\right]}{f} - \partial_x \frac{\operatorname{Log}\left[1+e^{i (e+f x)}\right]}{f}$

Rule: If $m \in \mathbb{Z}^+$, then

$$\int (c + d x)^m \operatorname{Sec}[e + f x] \, dx \longrightarrow$$

$$-\frac{2 i (c + d x)^m \operatorname{ArcTan}\left[e^{i (e+f x)}\right]}{f} - \frac{d m}{f} \int (c + d x)^{m-1} \operatorname{Log}\left[1 - i e^{i (e+f x)}\right] dx + \frac{d m}{f} \int (c + d x)^{m-1} \operatorname{Log}\left[1 + i e^{i (e+f x)}\right] dx$$

$$\int (c + d x)^m \operatorname{Csc}[e + f x] \, dx \longrightarrow$$

$$-\frac{2 (c + d x)^m \operatorname{ArcTanh}\left[e^{-i (e+f x)}\right]}{f} - \frac{d m}{f} \int (c + d x)^{m-1} \operatorname{Log}\left[1 - e^{-i (e+f x)}\right] dx + \frac{d m}{f} \int (c + d x)^{m-1} \operatorname{Log}\left[1 + e^{-i (e+f x)}\right] dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*csc[e_.+k_.*Pi+f_.*Complex[0,fz_]*x_],x_Symbol] :=
  -2*(c+d*x)^m*ArcTanh[E^(-I*k*Pi)*E^(-I*e+f*fz*x)]/(f*fz*I) -
  d*m/(f*fz*I)*Int[(c+d*x)^(m-1)*Log[1-E^(-I*k*Pi)*E^(-I*e+f*fz*x)],x] +
  d*m/(f*fz*I)*Int[(c+d*x)^(m-1)*Log[1+E^(-I*k*Pi)*E^(-I*e+f*fz*x)],x] /;
FreeQ[{c,d,e,f,fz},x] && IntegerQ[2*k] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*csc[e_.+k_.*Pi+f_.*x_],x_Symbol] :=
  -2*(c+d*x)^m*ArcTanh[E^(I*k*Pi)*E^(I*(e+f*x))]/f -
  d*m/f*Int[(c+d*x)^(m-1)*Log[1-E^(I*k*Pi)*E^(I*(e+f*x))],x] +
  d*m/f*Int[(c+d*x)^(m-1)*Log[1+E^(I*k*Pi)*E^(I*(e+f*x))],x] /;
FreeQ[{c,d,e,f},x] && IntegerQ[2*k] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*csc[e_.+f_.*Complex[0,fz_]*x_],x_Symbol] :=
  -2*(c+d*x)^m*ArcTanh[E^(-I*e+f*fz*x)]/(f*fz*I) -
  d*m/(f*fz*I)*Int[(c+d*x)^(m-1)*Log[1-E^(-I*e+f*fz*x)],x] +
  d*m/(f*fz*I)*Int[(c+d*x)^(m-1)*Log[1+E^(-I*e+f*fz*x)],x] /;
FreeQ[{c,d,e,f,fz},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*csc[e_.+f_.*x_],x_Symbol] :=
  -2*(c+d*x)^m*ArcTanh[E^(I*(e+f*x))]/f -
  d*m/f*Int[(c+d*x)^(m-1)*Log[1-E^(I*(e+f*x))],x] +
  d*m/f*Int[(c+d*x)^(m-1)*Log[1+E^(I*(e+f*x))],x] /;
FreeQ[{c,d,e,f},x] && IGtQ[m,0]
```

2. $\int (c+d\,x)^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $n>1$

1: $\int (c+d\,x)^m\,\mathrm{Sec}[e+f\,x]^2\,dx$ when $m>0$

Reference: CRC 430, A&S 4.3.125

Reference: CRC 428, A&S 4.3.121

Basis: $\mathrm{Sec}[e+f\,x]^2 == \partial_x \frac{\mathrm{Tan}[e+f\,x]}{f}$

–

Rule: If $m>0$, then

$$\int (c+d\,x)^m\,\mathrm{Sec}[e+f\,x]^2\,dx \longrightarrow \frac{(c+d\,x)^m\,\mathrm{Tan}[e+f\,x]}{f}-\frac{d\,m}{f}\int (c+d\,x)^{m-1}\,\mathrm{Tan}[e+f\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*csc[e_.+f_.*x_]^2,x_Symbol] :=
  -(c+d*x)^m*Cot[e+f*x]/f +
  d*m/f*Int[(c+d*x)^(m-1)*Cot[e+f*x],x] /;
FreeQ[{c,d,e,f},x] && GtQ[m,0]
```

2: $\int (c+d\,x)\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $n>1\land n\neq 2$

Reference: G&R 2.643.2 with $m\to 1$, CRC 431, A&S 4.3.126

Reference: G&R 2.643.1 with $m\to 1$, CRC 429', A&S 4.3.122

Rule: If $n>1\land n\neq 2$, then

$$\int (c+d\,x)\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx\longrightarrow$$
$$\frac{b^2\,(c+d\,x)\,\mathrm{Tan}[e+f\,x]\,(b\,\mathrm{Sec}[e+f\,x])^{n-2}}{f\,(n-1)}-\frac{b^2\,d\,(b\,\mathrm{Sec}[e+f\,x])^{n-2}}{f^2\,(n-1)\,(n-2)}+\frac{b^2\,(n-2)}{n-1}\int (c+d\,x)\,(b\,\mathrm{Sec}[e+f\,x])^{n-2}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)*(b_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^2*(c+d*x)*Cot[e+f*x]*(b*Csc[e+f*x])^(n-2)/(f*(n-1)) -
  b^2*d*(b*Csc[e+f*x])^(n-2)/(f^2*(n-1)*(n-2)) +
  b^2*(n-2)/(n-1)*Int[(c+d*x)*(b*Csc[e+f*x])^(n-2),x] /;
FreeQ[{b,c,d,e,f},x] && GtQ[n,1] && NeQ[n,2]
```

3: $\int (c+d\,x)^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $n>1\land n\neq 2\land m>1$

Reference: G&R 2.643.2

Reference: G&R 2.643.1

Rule: If $n>1\land n\neq 2\land m>1$, then

$$\int (c+d\,x)^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx\longrightarrow$$
$$\frac{b^2\,(c+d\,x)^m\,\mathrm{Tan}[e+f\,x]\,(b\,\mathrm{Sec}[e+f\,x])^{n-2}}{f\,(n-1)}-\frac{b^2\,d\,m\,(c+d\,x)^{m-1}\,(b\,\mathrm{Sec}[e+f\,x])^{n-2}}{f^2\,(n-1)\,(n-2)}+$$

$$\frac{b^2 (n-2)}{n-1} \int (c+d x)^m (b \operatorname{Sec}[e+f x])^{n-2} \, dx + \frac{b^2 d^2 m (m-1)}{f^2 (n-1) (n-2)} \int (c+d x)^{m-2} (b \operatorname{Sec}[e+f x])^{n-2} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_*(b_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -b^2*(c+d*x)^m*Cot[e+f*x]*(b*Csc[e+f*x])^(n-2)/(f*(n-1)) -
  b^2*d*m*(c+d*x)^(m-1)*(b*Csc[e+f*x])^(n-2)/(f^2*(n-1)*(n-2)) +
  b^2*(n-2)/(n-1)*Int[(c+d*x)^m*(b*Csc[e+f*x])^(n-2),x] +
  b^2*d^2*m*(m-1)/(f^2*(n-1)*(n-2))*Int[(c+d*x)^(m-2)*(b*Csc[e+f*x])^(n-2),x] /;
FreeQ[{b,c,d,e,f},x] && GtQ[n,1] && NeQ[n,2] && GtQ[m,1]
```

2. $\int (c+d x)^m (b \operatorname{Sec}[e+f x])^n \, dx$ when $n < -1$

1: $\int (c+d x) (b \operatorname{Sec}[e+f x])^n \, dx$ when $n < -1$

Reference: G&R 2.631.3 with $m \to 1$

Reference: G&R 2.631.2 with $m \to 1$

Rule: If $n < -1$, then

$$\int (c+d x) (b \operatorname{Sec}[e+f x])^n \, dx \longrightarrow$$

$$\frac{d (b \operatorname{Sec}[e+f x])^n}{f^2 n^2} - \frac{(c+d x) \operatorname{Sin}[e+f x] (b \operatorname{Sec}[e+f x])^{n+1}}{b f n} + \frac{n+1}{b^2 n} \int (c+d x) (b \operatorname{Sec}[e+f x])^{n+2} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)*(b_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  d*(b*Csc[e+f*x])^n/(f^2*n^2) +
  (c+d*x)*Cos[e+f*x]*(b*Csc[e+f*x])^(n+1)/(b*f*n) +
  (n+1)/(b^2*n)*Int[(c+d*x)*(b*Csc[e+f*x])^(n+2),x] /;
FreeQ[{b,c,d,e,f},x] && LtQ[n,-1]
```

2: $\int (c+d\,x)^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $n<-1\land m>1$

Reference: G&R 2.631.3

Reference: G&R 2.631.2

Rule: If $n<-1\land m>1$, then

$$\int (c+d\,x)^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx\;\longrightarrow$$
$$\frac{d\,m\,(c+d\,x)^{m-1}\,(b\,\mathrm{Sec}[e+f\,x])^n}{f^2\,n^2}-\frac{(c+d\,x)^m\,\mathrm{Sin}[e+f\,x]\,(b\,\mathrm{Sec}[e+f\,x])^{n+1}}{b\,f\,n}+$$
$$\frac{n+1}{b^2\,n}\int (c+d\,x)^m\,(b\,\mathrm{Sec}[e+f\,x])^{n+2}\,dx-\frac{d^2\,m\,(m-1)}{f^2\,n^2}\int (c+d\,x)^{m-2}\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_*(b_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  d*m*(c+d*x)^(m-1)*(b*Csc[e+f*x])^n/(f^2*n^2) +
  (c+d*x)^m*Cos[e+f*x]*(b*Csc[e+f*x])^(n+1)/(b*f*n) +
  (n+1)/(b^2*n)*Int[(c+d*x)^m*(b*Csc[e+f*x])^(n+2),x] -
  d^2*m*(m-1)/(f^2*n^2)*Int[(c+d*x)^(m-2)*(b*Csc[e+f*x])^n,x] /;
FreeQ[{b,c,d,e,f},x] && LtQ[n,-1] && GtQ[m,1]
```

3: $\int (c+d\,x)^m\,(b\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((b\,\mathrm{Cos}[e+f\,x])^n\,(b\,\mathrm{Sec}[e+f\,x])^n\right)==0$

Rule: If $n\notin\mathbb{Z}$, then

$$\int (c+d\,x)^m\,(b\,\text{Sec}[e+f\,x])^n\,dx \,\longrightarrow\, (b\,\text{Cos}[e+f\,x])^n\,(b\,\text{Sec}[e+f\,x])^n \int \frac{(c+d\,x)^m}{(b\,\text{Cos}[e+f\,x])^n}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(b_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  (b*Sin[e+f*x])^n*(b*Csc[e+f*x])^n*Int[(c+d*x)^m/(b*Sin[e+f*x])^n,x] /;
FreeQ[{b,c,d,e,f,m,n},x] && Not[IntegerQ[n]]
```

2: $\int (c+d\,x)^m\,(a+b\,\text{Sec}[e+f\,x])^n\,dx$ when $(m \mid n) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int (c+d\,x)^m\,(a+b\,\text{Sec}[e+f\,x])^n\,dx \,\longrightarrow\, \int (c+d\,x)^m\,\text{ExpandIntegrand}[(a+b\,\text{Sec}[e+f\,x])^n,\,x]\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m,(a+b*Csc[e+f*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[m,0] && IGtQ[n,0]
```

3: $\int (c+d\,x)^m\,(a+b\,\text{Sec}[e+f\,x])^n\,dx$ when $n \in \mathbb{Z}^- \land m \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: If $n \in \mathbb{Z}$, then $(a+b\,\text{Sec}[z])^n == \frac{\text{Cos}[z]^{-n}}{(b+a\,\text{Cos}[z])^{-n}}$

–

Rule: If $n \in \mathbb{Z}^- \land m \in \mathbb{Z}^+$, then

$$\int (c+d\,x)^m\,(a+b\,\text{Sec}[e+f\,x])^n\,dx \longrightarrow \int (c+d\,x)^m\,\text{ExpandIntegrand}\left[\frac{\text{Cos}[e+f\,x]^{-n}}{(b+a\,\text{Cos}[e+f\,x])^{-n}},\,x\right]dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m,Sin[e+f*x]^(-n)/(b+a*Sin[e+f*x])^(-n),x],x] /;
FreeQ[{a,b,c,d,e,f},x] && ILtQ[n,0] && IGtQ[m,0]
```

X: $\int (c+d\,x)^m\,(a+b\,\text{Sec}[e+f\,x])^n\,dx$

Basis: $\text{Csc}[e+f\,x] == \text{Sec}\left[e-\frac{\pi}{2}+f\,x\right]$

Basis: $\text{Csc}[e+f\,x] == i\,\text{Csch}[i\,e+i\,f\,x]$

Basis: $\text{Csc}[e+f\,x] == \text{Sech}\left[i\left(e-\frac{\pi}{2}\right)+i\,f\,x\right]$

Rule:

$$\int (c+d\,x)^m\,(a+b\,\text{Sec}[e+f\,x])^n\,dx \longrightarrow \int (c+d\,x)^m\,(a+b\,\text{Sec}[e+f\,x])^n\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*csc[e_.+f_.*x_]^n_.,x_Symbol] :=
  If[MatchQ[f,f1_.*Complex[0,j_]],
    If[MatchQ[e,e1_.+Pi/2],
      Unintegrable[(c+d*x)^m*Sech[I*(e-Pi/2)+I*f*x]^n,x],
    (-I)^n*Unintegrable[(c+d*x)^m*Csch[-I*e-I*f*x]^n,x]],
  If[MatchQ[e,e1_.+Pi/2],
    Unintegrable[(c+d*x)^m*Sec[e-Pi/2+f*x]^n,x],
  Unintegrable[(c+d*x)^m*Csc[e+f*x]^n,x]]] /;
FreeQ[{c,d,e,f,m,n},x] && IntegerQ[n]
```

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*(a+b*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

N: $\int u^m\,(a+b\,\mathrm{Sec}[v])^n\,dx$ when $u == c+d\,x\,\land\,v == e+f\,x$

Derivation: Algebraic normalization

Rule: If $u == c+d\,x\,\land\,v == e+f\,x$, then

$$\int u^m\,(a+b\,\mathrm{Sec}[v])^n\,dx\,\longrightarrow\,\int (c+d\,x)^m\,\left(a+b\,\mathrm{Sec}[e+f\,x]\right)^n\,dx$$

Program code:

```
Int[u_^m_.*(a_.+b_.*Sec[v_])^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*(a+b*Sec[ExpandToSum[v,x]])^n,x] /;
FreeQ[{a,b,m,n},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

```
Int[u_^m_.*(a_.+b_.*Csc[v_])^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*(a+b*Csc[ExpandToSum[v,x]])^n,x] /;
FreeQ[{a,b,m,n},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```